/*-------------------------------------------------------------------+

|                                                                    +

|   This program saves all user libraries and can be easily          +

|   modified to save others.                                         +

|                                                                    +

\-------------------------------------------------------------------*/

   PGM        PARM(&DEV)

             DCL        VAR(&DEV) TYPE(*CHAR) LEN(10) /* Only one +

                          device */

             DCLF       FILE(QADSPOBJ)

             DCL        VAR(&QCTLSBS) TYPE(*CHAR) LEN(20)

             DCL        VAR(&KEY) TYPE(*CHAR) LEN(4)

             DCL        VAR(&REPLY) TYPE(*CHAR) LEN(1)

  /*  Check with user before going further. */

             SNDPGMMSG  MSG('This program will end the system to a +

                          restricted state. Continue (Y or N)?') +

                          TOPGMQ(*EXT) MSGTYPE(*INQ) KEYVAR(&KEY)

             RCVMSG     PGMQ(*SAME) MSGTYPE(*RPY) MSGKEY(&KEY) +

                          WAIT(*MAX) RMV(*NO) MSG(&REPLY)

             IF         COND(&REPLY *NE 'Y' *AND &REPLY *NE 'y') +

                          THEN(RETURN)

             CRTLIB     LIB(SAVLIB) TYPE(*PROD)

             MONMSG     MSGID(CPF0000) EXEC(CLRLIB LIB(SAVLIB))

             DSPOBJD    OBJ(*ALL) OBJTYPE(*LIB) OUTPUT(*OUTFILE) +

                          OUTFILE(SAVLIB/ZZZZZZLIBS)

             RTVSYSVAL  SYSVAL(QCTLSBSD) RTNVAR(&QCTLSBS)

             ENDSYS     DELAY(300)

  /*  Loop thru list for each library, except SAVLIB and QTEMP  */

             OVRDBF     FILE(QADSPOBJ) TOFILE(SAVLIB/ZZZZZZLIBS)

 LOOP1:      RCVF

             MONMSG     MSGID(CPF0864) EXEC(GOTO CMDLBL(ENDLOOP1))

           /*  Checks for system libraries.  */

             IF         COND(%SST(&ODOBNM 1 1) = 'Q') THEN(DO)

               /*  Libraries that start with Q are system libraries  +

                   except QGPL, QUSER*, & QUSR*. Note, there are a   +

                   few more but this program is not going to save    +

                   them.                                            */

               IF         COND(&ODOBNM = 'QGPL') THEN(GOTO CMDLBL(SAVE))

               IF         COND(%SST(&ODOBNM 1 5) = 'QUSER') THEN(GOTO +

                            CMDLBL(SAVE))

               IF         COND(%SST(&ODOBNM 1 4) = 'QUSR') THEN(GOTO +

                            CMDLBL(SAVE))

               GOTO       CMDLBL(LOOP1)

             ENDDO

           /*  Check for work libraries. */

             IF         COND(&ODOBNM = 'SAVLIB' *OR &ODOBNM = +

                          'QTEMP') THEN(GOTO CMDLBL(LOOP1))

 SAVE:       CRTSAVF    FILE(SAVLIB/&ODOBNM) MAXRCDS(*NOMAX)

             SAVLIB     LIB(&ODOBNM) DEV(*SAVF) SAVF(SAVLIB/&ODOBNM) +

                          DTACPR(*YES)

             GOTO       CMDLBL(LOOP1)

 ENDLOOP1:

             STRSBS     SBSD(%SST(&QCTLSBS 11 10)/%SST(&QCTLSBS 1 10))

  /*  Spawn-off batch job that will save the save files to tape      +

      once the system has come back up.                              +

 */

             SBMJOB     CMD(CALL PGM(SAVELIBS2) PARM(&DEV)) +

                          JOB(SAVELIBS2)

  Return




